
This exam (like the previous two and the next three) is in two parts: first 75 minutes here, written, closed 
book and individual work. Then, in lab, still individual work but you can check the Internet, your notes, 
work with DrRacket and ask the Uis/Ais for help. In lab you are not allowed to add or remove text to your 
exam, just type your answer(s) in, exactly as they appear on the paper, check them, then fix them and 
submit a brief statement as to what your raw score should be. Your .rkt file should contain the answer(s) 
exactly as written below in comment boxes, the fixed code1 (if necessary to fix it, but make sure what you 
submit is in working condition) and then the grade you think your written exam deserves2.  

1. Design a function that takes a possibly empty list of numbers (integers) and sums them up. How much 
simpler is your solution if you use one of the fold functions. Does it get even simpler if you use ISL’s apply? 

; sum : [ListOf Number] -> Number 
; 
(define (sum lon) 
  (apply + lon)) 
 

Another possibility is this (a little longer, still very short): 
 
(define (sum lon) 
  (foldl + 0 lon)) 

 
Another possibility is this (basic structural recursion template): 

 
(define (sum lon) 
  (cond ((empty? lon) 0) 
        ((cons? lon) (+ (first lon) (sum (rest lon)))) 
        (else "Bad data."))) 

 
Here’s how we test either one of these functions:  
 

(check-expect (sum empty) 0) 
(check-expect (sum (build-list 100 add1)) 
              (/ (* 100 (add1 100)) 2)) 

2. Design a function that takes a possibly empty list of numbers (integers) and determines and reports the 
largest number in the list. If the input list is empty the result should be false. Don’t forget the design 
recipe, at the very least provide the signature of the function. What’s the return type of your function?  

; largest : [ListOf Number] -> [Maybe Number] 
; 
(define (largest lon) 
  (cond ((empty? lon) false) 
        ( else (apply max lon)))) 

Note the signature: the return type is a parametric type.  
 

                                                           
1 Fixes should be the cheapest (simplest) you can think of that preserve your initial intent. 
2 There are eight problems in this exam. Each weights the same. I need a percentage for each, then their average. 

Here are some tests: 
 

(check-expect (largest empty) false) 
(check-expect (largest (list -3)) -3) 
(check-expect (largest (list 4 7 2 3 7 3 1)) 7) 

 
Another solution would be like this (includes definition of the said parametric type):  
 

; [Maybe X] is one of: 
;   -- X 
;   -- false 
 
; largest : [ListOf Number] -> [Maybe Number] 
; 
(define (largest lon) 
  (cond ((empty? lon) false) 
        ((cons? lon) (let ((v (largest (rest lon)))) 
                       (if (false? v) 
                         (first lon) 
                         (if (> (first lon) v)  
                             (first lon) 
                             v))))  
        (else (error "Bad input.")))) 
 
(check-expect (largest empty) false) 
(check-expect (largest (list 3)) 3) 
(check-expect (largest empty) false) 
(check-expect (largest (list -3)) -3) 
(check-expect (largest (list 4 7 2 3 7 3 1)) 7) 
 

If you don’t want to use ISL (and local or let) you have: 

(define (largest lon) 
  (cond ((empty? lon) false) 
        ((cons? lon) (if (false? (largest (rest lon))) 
                         (first lon) 
                         (if (> (first lon) (largest (rest lon))) 
                             (first lon) 
                             (largest (rest lon))))) 
        (else (error "Bad input.")))) 
 
(define a (list 5 2 3 7 2 -1 -12 16 3 0 0 2 11)) 
 
(check-expect (largest a) 16) ; see a above 

3. Design a function that takes a possibly empty list of numbers (integers) and identifies and removes from 
the output the even numbers. The result is a new list with the odd numbers from the input in the same 
order. How much simpler does your function become if you decide to use ISL’s filter? 

; signature below 
; 
(define (remove-even lon) 
  (filter odd? lon)) 
 
(check-expect 
 (remove-even (list 1 2 3 4 5 6 7 8 9)) 
              (list 1 3 5 7 9)) 

If we want to practice the structural recursion template a bit more we have: 

; remove-even : [ListOf Number] -> [ListOf Number] 
; 
(define (remove-even lon) 
  (cond ((empty? lon) lon) 
        ((cons? lon) (if (even? (first lon)) 
                         (remove-even (rest lon)) 
                         (cons (first lon) 
                               (remove-even (rest lon))))) 
        (else (error "Bad input.")))) 
 
(check-expect (remove-even (list 1 2 3 4 5 6 7 8 9)) 
              (list 1 3 5 7 9)) 
 
(check-expect (remove-even empty) empty) 

4. Design a function that receives a possibly empty list of numbers (integers)  and returns a new list with 
the same numbers as in its input but sorted such that all the negative numbers (if any) are listed first, in 
descending order, followed by the positive integers (if any) in ascending order. Any 0’s (if there are any) 
should be at the very end in the sorted result. 

; six : [ListOf Number] -> [ListOf Number] 
; 
(define (six lon) 
  (sort lon (lambda (a b) 
              (cond ((and (< a 0) (< b 0)) (> a b)) 
                    ((and (> a 0) (> b 0)) (< a b)) 
                    ((and (< a 0) (> b 0)) true) 
                    ((and (> a 0) (< b 0)) false) 
                    ((zero? a) false) 
                    ((zero? b) true))))) 
                       
(check-expect (six  
                (list 0 1 0 -1 5 -2 0 2 -5 -1)) 
              (list -1 -1 -2 -5 1 2 5 0 0 0)) 

(six (build-list  10 ; how many  
                 (lambda (x)  
                   (- (random 6) ; scaling 
                       3))))     ; shifting (translating) 
; this last case suggests an answer for the next problem 

 

5. Please design a function that produces a list of random integers. The function has only one argument 
(the size of the resulting list) and it should simulate throwing a dice (so the integers in the list should be 
numbers between 1 and 6 (inclusive)). Don’t forget the design recipe. How much simpler does your answer 
become if you decide to use the build-list primitive that ISL provides. 

; A Dice is one of: 
;   -- 1 
;   -- 2 
;   -- 3 
;   -- 4 
;   -- 5 
;   -- 6 
 
; dice : Number -> [ListOf Dice] 
; 
(define (dice size) 
  (build-list size 
              (lambda (x) (add1 (random 6))))) 

 

6. Define a function called extract that receives a possibly empty string of characters. The function 
randomly picks up one of the characters in the string and returns it. Assume you can use substring (the 
function, with either one or two arguments) but you cannot use string-ith. Also, you’re not allowed to 
turn the string into any other type (e.g., list of characters). You are definitely allowed to use random3.   

; extract : String -> String 
; takes one random character from the input and returns it 
(define (extract word) 
  (substring 
   (substring word (random (string-length word)))  
   ; slice this string randomly somewhere (at some index) 
   0 1) ; then get the first character of that slice  
 
(sort 
 (build-list 
  20 
  (lambda (x) 
    (extract "nothing"))) 
 string<?) ; sorted list of random chars from "nothing" 

                                                           
3 (random n) will produce a random integer between 0 and n – 1 (inclusive).  

Another way would be to define a helper: 
 

; extract : String -> String 
; takes one random character from the input and returns it 
(define (extract word) 
  (pick word (random (string-length word)))) 
 
; pick : String Number -> String 
; essentially we're implementing our own (string-ith word 1) 
(define (pick word index) 
  (substring word index (add1 index))) 

 

7. Is there a difference between 

(foldr - 1 (list 2 3 4 5 6)) and  

(foldl - 1 (list 2 3 4 5 6)) and if so what is it?  

No there is no difference.  

However if the lists had been one shorter there would have been. 

Consider the following: 

(check-expect (foldr / 1 (list 2 3 4 5)) 
              (/ 2 (/ 3 (/ 4 (/ 5 1))))) 
 
(check-expect (foldl / 1 (list 2 3 4 5)) 
              (/ 5 (/ 4 (/ 3 (/ 2 1))))) 
 
(check-expect (foldr - 1 (list 2 3 4 5)) 
              (- 2 (- 3 (- 4 (- 5 1))))) 
 
(check-expect (foldl - 1 (list 2 3 4 5)) 
              (- 5 (- 4 (- 3 (- 2 1))))) 

 
Now as one might expect these two expressions are different: 

 
(check-expect (foldl - 1 (list 2 3 4 5)) 3) 
 
(check-expect (foldr - 1 (list 2 3 4 5)) -1) 

 
However:  

 
(check-expect (foldl - 1 (list 2 3 4 5 6)) 
              (foldr - 1 (list 2 3 4 5 6))) 

8. What’s the value of this expression? 

(build-list 5 (lambda (x) (/ x 10))) 

Here’s one way to put it:  

(check-expect 
 (build-list 5 (lambda (x) (/ x 10))) 
 (map 
  (lambda (x) (/ x 10)) 
  (list 0 1 2 3 4))) 


